Distance et perpendiculaire commune de deux droites

On commence par charger le fichier contenant la définition de deux macros :

PointDeLaDroite – détermination d'un point d'une droite définie comme intersection de deux plans,

VecteurDeLaDroite – détermination d'un vecteur directeur d'une droite définie comme intersection de de deux plans.

Ce même fichier charge `vect.mac` qui contient la définition de ~ servant à calculer un produit vectoriel.

```
> load("geo3d.mc")$
```

On introduit la première droite $\mathscr{D}_1$.

```
> D1:[x=3*z+1,y=2*z-1];
```

$$\left[x=3\,z+1, y=2\,z-1\right]$$

Puis $\mathscr{D}_2$.

```
> D2:[y=x-2,z=1];
```

$$\left[y=x-2, z=1\right]$$

On détermine un point et un vecteur directeur de $\mathscr{D}_1$.

```
> block(A1:PointDeLaDroite(D1,z=0),V1:VecteurDeLaDroite(D1,z=0),[A1,V1]);
```

$$[[1,-1,0],[3,2,1]]$$

On en fait autant pour $\mathscr{D}_2$.

```
> block(A2:PointDeLaDroite(D2,x=0),V2:VecteurDeLaDroite(D2,x=0),[A2,V2]);
```

$$[[0,-2,1],[1,1,0]]$$

On détermine, à l'aide du produit vectoriel, un vecteur directeur de la perpendiculaire commune Δ.

```
> V:express(V1~V2);
```

$$[-1,1,1]$$

On introduit un point générique.

```
> M:[x,y,z];
```

$$[x,y,z]$$

On détermine une équation du plan contenant $\mathscr{D}_1$ et Δ.

```
> P1:expand((M-A1).express(V~V1))=0;
```

$$-5\,z+4\,y-x+5=0$$

Maintenant le plan contenant $\mathscr{D}_2$ et Δ.

```
> P2:expand((M-A2).express(V~V2))=0;
```

$$-2\,z+y-x+4=0$$

Nous pouvons définir Δ comme intersection des deux plans précédents.

```
> Delta:[P1,P2];
```

$$\left[-5\,z+4\,y-x+5=0,-2\,z+y-x+4=0\right]$$

À titre d'exemple voici un point de Δ.

```
> A:PointDeLaDroite(Delta,y=x-2);
```

$$\left[\frac{8}{3},\frac{2}{3},1\right]$$

Et la distance des deux droites $\mathscr{D}_1$ et $\mathscr{D}_2$.

```
> d:abs((A2-A1).V)/sqrt(V.V);
```

$$\frac{1}{\sqrt{3}}$$